High Load Monitor

When ALCS experiences a high work load, it is sometimes difficult for the user to find out what the high load consists of.

The high load monitor uses the monitor exit USRTIM2 to detect a high load situation and place information on the diagnostic file about each ECB in the system. Also there is an offline program which reads the diagnostic file and generates a simple report.

This package is suitable for both ALCS V2.2.1 and ALCS V2.3.1 systems.

There are three items you will need on your host:

HILOAD USRTIM2	Source code for the monitor exit USRTIM2
HILOAD IW0USR	Source code for the IW0USR macro used by USRTIM2
HILOAD OFFLINE	Source code for the offline report generator

Notes about the use of the High Load Monitor and the report it produces

The USRTIM2 monitor exit code checks system load at 1 second intervals and if a high load is experienced, places information on the diagnostic file (and a warning message on the system console). High load data is written to the diagnostic file at 15 second intervals while the high load condition remains.

This user exit is designed for use on systems with any number of CPU loop tasks but has only been tested on systems with one CPU loop task.

The High Load Monitor regards the system load as high if the total number of ECBs (active plus delay/defer) exceeds (ACV1*ACV6)/100.

HILOAD OFFLINE is the source code of an offline program. Load this to your program library and assemble and link it to generat, called HILOAD, that processes diagnostic files containing high load data and formats this data into a simple report.

The following sample JCL can be used for running the offline program.

```
//HILOAD EXEC PGM=HILOAD
//STEPLIB  DD DSNAME=MY.ALCS.PROGLIB,DISP=SHR
//FILEIN   DD DSNAME=DXC.V2R3M1.SEQF.DIA.M0001000,DISP=(OLD,KEEP)
//FILEOUT  DD SYSOUT=*
```

The following is sample output from the HILOAD report program.

```
*********************************************************************

DATE   TIME        ACT  DEL  SU   IOCB ACV1 ACV2 ACV3 ACV4 ACV5 ACV6
05.342 15.10.26    0013 0009 0012 01F6 0019 001D 000A 0019 0019 0046

CRI    ECB      PROG MACRO -HLD- AGE
020178 0F81F5A0 PGM        00 00 0001
020178 0F82F5A0 PGM  EXITC 00 00 0001
020178 0F84F5A0 LIFX CREMC 00 00 0001
020178 0F8575A0 LIFX CREMC 00 00 0001
```

The report header contains:

	Date/Time of high load condition
ACT	Active ECB count
DEL	Delay/Defer ECB count
SU	Available Type 1 Storage Unit count
IOCB	Available IOCB count
ACVn	Activity Control Variable n.

And for each ECB:

The originating terminal,
The program name,
The last macro,
The number of records held by the ECB,
The number of resources held by the ECB
The ECB age (in seconds),

All counts are shown in hexadecimal.

HILOAD USRTIM2

```
USRTIM2  TITLE ' *** USRTIM2 INSTALLATION-WIDE MONITOR EXIT *** '       00001000
*=====================================================================* 00002000
*        ALCS INSTALLATION-WIDE MONITOR EXIT USRTIM2                  * 00003000
*=====================================================================* 00004000
*                                                                     * 00005000
*        THE PURPOSE OF THIS EXIT IS TO ALLOW USER TIME RELATED       * 00006000
*        PROCESSING TO TAKE PLACE.                                    * 00007000
*        USE THIS EXIT FOR INSTANCE TO UPDATE USER TABLES.            * 00008000
*                                                                     * 00009000
*---------------------------------------------------------------------* 00010000
*                                                                     * 00011000
*        ALCS WILL CALL TIMER INTERRUPT HANDLING WILL CALL            * 00012000
*        THIS EXIT ABOUT EVERY SECOND.                                * 00013000
*                                                                     * 00014000
*---------------------------------------------------------------------* 00015000
         EJECT ,                                                       00016000
*---------------------------------------------------------------------* 00017000
*                                                                     * 00018000
*        THIS EXIT MUST NOT MODIFY REGISTERS R11 THROUGH R12.         * 00019000
*        THIS EXIT MUST RESTORE ALL REGISTERS EXCEPT R15.             * 00020000
*                                                                     * 00021000
*        ENTRY CONDITIONS:                                            * 00022000
*        -----------------                                            * 00023000
*                                                                     * 00024000
*        NO PARAMETERS ARE PASSED BY ALCS TO THIS EXIT                * 00025000
*                                                                     * 00026000
*        R12 - INSTALLATION-WIDE MONITOR EXIT INFORMATION             * 00027000
*              TABLE ADDRESS (IW0IT)                                  * 00028000
*                                                                     * 00029000
*        R13 - CALLING PROGRAM SAVE AREA ADDRESS                      * 00030000
*                                                                     * 00031000
*        R15 - ADDRESS OF THIS MODULE                                 * 00032000
*                                                                     * 00033000
*        EXIT CONDITIONS:                                             * 00034000
*        ----------------                                             * 00035000
*                                                                     * 00036000
*        ALL REGISTERS SAME AS ON ENTRY EXCEPT R15:                   * 00037000
*                                                                     * 00038000
*            R15 = 0  MUST BE SET STANDARD                            * 00039000
*                                                                     * 00040000
*=====================================================================* 00041000
         EJECT ,                                                       00042000
         DXCUHDR NAME=USRTIM2,ENTRY=UTIM2000                           00043000
         EJECT ,                                                       00044000
*=====================================================================* 00045000
*                                                                     * 00046000
*        USER HEADER COMMENTS SECTION STARTS HERE                     * 00047000
*                                                                     * 00048000
*---------------------------------------------------------------------* 00049000
****                                                                  * 00050000
**** THIS USRTIM2 CONTAINS THE HIGH LOAD MONITOR ROUTINE        HILOAD* 00051000
****                                                                  * 00052000
**** UPDATED IN JAN 2001 TO USE CORRECT BASE FOR "OUTPUT AREA"  FIX001* 00052100
****                                                                  * 00052200
*---------------------------------------------------------------------* 00053000
*                                                                     * 00054000
*        USER HEADER COMMENTS SECTION ENDS HERE                       * 00055000
*                                                                     * 00056000
*=====================================================================* 00057000
         EJECT ,                                                       00058000
```

```
*=====================================================================* 00059000
*                                                                     * 00060000
*        ALCS MACRO DEFINITIONS                                       * 00061000
*                                                                     * 00062000
*=====================================================================* 00063000
         SPACE 1                                                      00064000
         MN0SV REG=R13             MONITOR SAVE AREA DSECT            00065000
         PRINT NOGEN                                                  00066000
         SPACE 1                                                      00067000
MN0SV    DSECT ,                   CONTINUE WORK AREA DSECT           00068000
         SPACE 1                                                      00069000
PARMLIST DC    8F'0'               CALLABLE SERVICES PARAMETER LIST   00070000
         SPACE 1                                                      00071000
*                                  USER SAVE AREA EXTENSIONS GO HERE  00072000
STORSIZE DC    F'0'                SAVE AREA FOR STRGE VALIDATE  HILOAD 00073000
STORIND  DC    X'0'                SAVE AREA FOR FILE IND        HILOAD 00074000
*                                                                    00075000
*                                  END OF USER SAVE AREA EXTENSIONS   00076000
         SPACE 1                                                      00077000
         DC    0D'0'               FORCE LENGTH TO DOUBLEWORDS        00078000
         SPACE 1                                                      00079000
USRSAV   EQU   *-MN0SV                                               00080000
USRTIM2  CSECT ,                   RESTORE CSECT                      00081000
         SPACE 1                                                      00082000
PARMS    DSECT ,                   DSECT FOR PARAMETER LIST           00083000
PARM1    DS    F                   PARAMETER 1                        00084000
PARM2    DS    F                   PARAMETER 2                        00085000
PARM3    DS    F                   PARAMETER 3                        00086000
PARM4    DS    F                   PARAMETER 4                        00087000
PARM5    DS    F                   PARAMETER 5                        00088000
PARM6    DS    F                   PARAMETER 6                        00089000
PARM7    DS    F                   PARAMETER 7                        00090000
PARM8    DS    F                   PARAMETER 8                        00091000
         USING PARMS,R01                                              00092000
USRTIM2  CSECT ,                   RESTORE CSECT                      00093000
         SPACE 1                                                      00094000
*=====================================================================* 00095000
*                                                                     * 00096000
*        END OF ALCS MACRO DEFINITIONS                                * 00097000
*                                                                     * 00098000
*=====================================================================* 00099000
         EJECT ,                                                      00100000
*=====================================================================* 00101000
*                                                                     * 00102000
*        USER MACRO DEFINITIONS SHOULD BE INSERTED BELOW              * 00103000
*                                                                     * 00104000
*=====================================================================* 00105000
         SPACE 1                                                      00106000
         IW0USR  ,                 IW0IT USER AREA DEFINITIONS HILOAD 00107000
         LS0LS REG=0                                          HILOAD 00108000
         CP0DA REG=R14                                        HILOAD 00109000
         SPACE 1                                                      00110000
*=====================================================================* 00111000
*                                                                     * 00112000
*        END OF USER MACRO DEFINITIONS                                * 00113000
*                                                                     * 00114000
*=====================================================================* 00115000
         PRINT NOGEN                                                  00116000
USRTIM2  CSECT ,                   RESTORE CSECT                      00117000
         EJECT ,                                                      00118000
```

```
*=====================================================================* 00119000
*                                                                     * 00120000
*        USER OBJECT COMMENTS SHOULD BE INSERTED BELOW                * 00121000
*                                                                     * 00122000
*=====================================================================* 00123000
         SPACE 1                                                      00124000
         DC    CL8'  HILOAD'                                          00125000
         SPACE 1                                                      00126000
*=====================================================================* 00127000
*                                                                     * 00128000
*        END OF USER OBJECT COMMENTS                                  * 00129000
*                                                                     * 00130000
*=====================================================================* 00131000
         EJECT ,                                                      00132000
UTIM2000 DC    0H'0'                                                  00133000
         DXCSAVE SAVE,PUSH,        SAVE CALLERS REGISTERS             -00134000
               ID=USER,            AND CHAIN THE USER SAVE AREA       -00135000
               SIZE=512,           LENGTH DEFINED IN SAVE AREA        -00136000
               WORKREG=R02                                             00137000
         SPACE 1                                                      00138000
         DROP  R15                 DROP PREVIOUS BASE REGISTER        00139000
         USING USRTIM2,R10                                            00140000
         LR    R10,R15             USE R10 AS BASE REGISTER           00141000
         SR    R15,R15             SET GOOD RETURN CODE               00142000
         SPACE 1                                                      00143000
*=====================================================================* 00144000
*                                                                     * 00145000
*        START OF USER CODE                                           * 00146000
*                                                                     * 00147000
*=====================================================================* 00148000
         SPACE 1                                                      00149000
*-------------------------------------------------------------  HILOAD 00150000
*        IMPLEMENTATION OF HIGH LOAD MONITOR                   HILOAD 00151000
*-------------------------------------------------------------  HILOAD 00152000
         SPACE 2                                               HILOAD 00153000
*-------------------------------------------------------------  HILOAD 00154000
*        CHECK FOR HIGH LOAD CONDITION                         HILOAD 00155000
*-------------------------------------------------------------  HILOAD 00156000
         SPACE 1                                               HILOAD 00157000
******** FIRST CHECK IF WE ARE IN NORM STATE                    HILOAD 00158000
         SPACE 1                                               HILOAD 00159000
         L     R14,IW0DATAB                                    HILOAD 00160000
         TM    CP0STI,CCNORM       ARE WE IN NORM STATE        HILOAD 00161000
         BZ    UTIM2800            NO - BRANCH  RETURN         HILOAD 00162000
         SPACE 1                                               HILOAD 00163000
         L     R14,IW0ECBCT        ADDRESS OF ECB COUNT        HILOAD 00164000
         L     R14,0(,R14)         ECB COUNT                   HILOAD 00165000
         SRDL  R14,16              ACTIVE ECB COUNT IN R14     HILOAD 00166000
         SRL   R15,16              DELAY/DEFER COUNT IN R15    HILOAD 00167000
         AR    R14,R15             TOTAL ECB COUNT IN R14      HILOAD 00168000
         SPACE 1                                               HILOAD 00169000
         CS0DT REG=R15             SYSTEM TABLE DSECT          HILOAD 00170000
         SPACE 1                                               HILOAD 00171000
         L     R15,IW0CDS          ADDRESS OF SYSTEM TABLE     HILOAD 00172000
         L     R03,CS0ACV1                                     HILOAD 00173000
         MH    R03,CS0ACV6+2                                   HILOAD 00174000
         SR    R02,R02             CLEAR FOR DIVIDE            HILOAD 00175000
         D     R02,=F'100'         QUOTIENT IN R03             HILOAD 00176000
         CR    R14,R03             ECB COUNT > ACV1 * ACV6     HILOAD 00177000
         BNH   UTIM2800            NO - RETURN                 HILOAD 00178000
         SPACE 1                                               HILOAD 00179000
******** CHECK IF IT IS 15 SECONDS SINCE THE LAST REPORT        HILOAD 00180000
         SPACE 1                                               HILOAD 00181000
```

```
         STCK  UTIMTNOW+USRSAV(R13) TIME NOW                      FIX001 00182000
         L     R14,IW0HITOD              TIME LAST REPORT       HILOAD 00183000
         LTR   R14,R14             IS THIS FIRST REPORT         HILOAD 00184000
         BZ    REPORT              YES - BRANCH                 HILOAD 00185000
         SPACE 1                                                HILOAD 00186000
         AH    R14,=H'15'          TIME LAST REPORT + 15        HILOAD 00187000
         C     R14,UTIMTNOW+USRSAV(,R13) TIME FOR NEW REPORT    FIX001 00188000
         BH    UTIM2800            NO - BRANCH RETURN           HILOAD 00189000
         EJECT ,                                                HILOAD 00190000
*--------------------------------------------------------------  HILOAD 00191000
*        STORE INFORMATION IN USRTIM SAVE AREA                  HILOAD 00192000
*--------------------------------------------------------------  HILOAD 00193000
         SPACE 1                                                HILOAD 00194000
******** LAYOUT OF HEADER                                       HILOAD 00195000
         SPACE 1                                                HILOAD 00196000
UTIMTNOW EQU   0                   TIME NOW                     HILOAD 00197000
UTIMTLAS EQU   8                   TIME LAST REPORT             HILOAD 00198000
*                                  (THESE 16 BYTES ARE A WORK   HILOAD 00199000
*                                  (AREA NOT CLEARED AND NOT    HILOAD 00200000
*                                  (DUMPED                      HILOAD 00201000
UTIMID   EQU   16                  BLOCK IDENTIFIER             HILOAD 00202000
*                                  EIP - PRIME BLOCK            HILOAD 00203000
*                                  EIO - OVERFLOW BLOCK         HILOAD 00204000
UTIMACT  EQU   20                  ACTIVE ECB COUNT             HILOAD 00205000
UTIMDEL  EQU   22                  DELAY/DEFER ECB COUNT        HILOAD 00206000
UTIMSUS  EQU   24                  AVAILABLE SUS                HILOAD 00207000
UTIMIOS  EQU   28                  AVAILABLE IOCBS              HILOAD 00208000
UTIMACV1 EQU   32                  ACV1                         HILOAD 00209000
UTIMACV2 EQU   36                  ACV2                         HILOAD 00210000
UTIMACV3 EQU   40                  ACV3                         HILOAD 00211000
UTIMACV4 EQU   44                  ACV4                         HILOAD 00212000
UTIMACV5 EQU   48                  ACV5                         HILOAD 00213000
UTIMACV6 EQU   52                  ACV6                         HILOAD 00214000
UTIMTIM  EQU   56                  TIME DATA DUMPED             HILOAD 00215000
UTIMHDRL EQU   48                  LENGTH OF HEADER TO DUMP     HILOAD 00216000
UTIMITM1 EQU   64                  FIRST ITEM (OF 18)           HILOAD 00217000
UTIMITML EQU   20                  ITEM LENGTH                  HILOAD 00218000
UTIMITMC EQU   18                  ITEM COUNT                   HILOAD 00219000
         SPACE 1                                                HILOAD 00220000
******** LAYOUT OF ITEM                                        HILOAD 00221000
         SPACE 1                                                HILOAD 00222000
ITMPFL   EQU   0                   BLOCK USAGE    (LS0PFL)      HILOAD 00223000
ITMCRI   EQU   1                   CRI            (CE3CRI)      HILOAD 00224000
ITMMAC   EQU   4                   LAST MACRO                   HILOAD 00225000
ITMRHN   EQU   6                   RECORDS HELD   (CE3RHN)      HILOAD 00226000
ITMCHN   EQU   7                   RESOURCES HELD (CE3CHN)      HILOAD 00227000
ITMECB   EQU   8                   ECB ADDR                     HILOAD 00228000
ITMPRG   EQU   12                  PROGRAM NAME                 HILOAD 00229000
ITMTOD   EQU   16                  DISPENSE TIME  (LS0TOD)      HILOAD 00230000
         EJECT ,                                                HILOAD 00231000
******** FIRST CLEAR OUTPUT AREA                               HILOAD 00232000
         SPACE 1                                                HILOAD 00233000
REPORT   EQU   *                                                HILOAD 00234000
         WTO   MF=(E,OUTMSG)       SEND WARNING MSG TO OPERATOR HILOAD 00235000
         SPACE 1                                                HILOAD 00236000
         MVC   IW0HITOD,UTIMTNOW+USRSAV(R13)                    FIX001-00237000
                                   UPDATE TIME LAST REPORT      HILOAD 00238000
         XC    USRSAV+16(256-16,R13),USRSAV+16(R13)             FIX001 00239000
         XC    USRSAV+256(256-USRSAV,R13),USRSAV+256(R13)       FIX001 00240000
         SPACE 1                                                HILOAD 00241000
         MVC   UTIMID+USRSAV(3,R13),=C'EIP' PRIME BLOCK         FIX001 00242000
         SPACE 1                                                HILOAD 00243000
         L     R14,IW0ECBCT        ADDRESS OF ECB COUNT         HILOAD 00244000
```

```
         MVC   UTIMACT+USRSAV(4,R13),0(R14) ECB COUNTS           FIX001 00245000
         SPACE 1                                                HILOAD 00246000
         L     R14,IW0SUNDX                                     HILOAD 00247000
         L     R14,LS0NBR-LS0LS(R14)                            HILOAD 00248000
         ST    R14,UTIMSUS+USRSAV(,R13) AVAILABLE SUS           FIX001 00249000
         SPACE 1                                                HILOAD 00250000
         L     R14,IW0IONDX                                     HILOAD 00251000
         L     R14,LS0NBR-LS0LS(R14)                            HILOAD 00252000
         ST    R14,UTIMIOS+USRSAV(,R13) AVAILABLE IOCBS         FIX001 00253000
         SPACE 1                                                HILOAD 00254000
         L     R15,IW0CDS          LOAD ADDRESS OF SYSTEM TABLE HILOAD 00255000
         MVC   UTIMACV1+USRSAV(24,R13),CS0ACV1 ACVS (6)         FIX001 00256000
         SPACE 1                                                HILOAD 00257000
         MVC   UTIMTIM+USRSAV(8,R13),UTIMTNOW+USRSAV(R13)       FIX001-00258000
                               TIME THIS REPORT                 HILOAD 00259000
         SPACE 1                                                HILOAD 00260000
         L     R14,IW0SUNDX                                     HILOAD 00261000
         L     RGF,LS0TOT-LS0LS(R14) TOTAL SUS (LOOP COUNT)     HILOAD 00262000
         LA    R06,LS0PTR-LS0LS(R14) POINT TO FIRST DESCRIPTOR  HILOAD 00263000
         SPACE 1                                                HILOAD 00264000
******** LOOP TO HERE FOR NEW OUTPUT BLOCK                      HILOAD 00265000
         SPACE 1                                                HILOAD 00266000
LOOP1    EQU   *                                                HILOAD 00267000
         LA    R05,UTIMITMC        ITEM COUNT (LOOP COUNT)      HILOAD 00268000
         LA    R04,UTIMITM1+USRSAV(,R13) POINT TO FIRST ITEM    FIX001 00269000
         SPACE 1                                                HILOAD 00270000
******** LOOP TO HERE FOR NEXT ITEM IN OUTPUT BLOCK             HILOAD 00271000
         SPACE 1                                                HILOAD 00272000
LOOP2    EQU   *                                                HILOAD 00273000
         MVC   ITMTOD(4,R04),LS0TOD-LS0PTR(R06) MOVE TOD        HILOAD 00274000
         MVC   ITMPFL(1,R04),LS0PFL-LS0PTR(R06) MOVE USAGE BYTE HILOAD 00275000
         TM    LS0PFL-LS0PTR(R06),LS0PFLI ITEM IN USE           HILOAD 00276000
         BNO   NEXT                NO - BRANCH                  HILOAD 00277000
         SPACE 1                                                HILOAD 00278000
         TM    LS0PFL-LS0PTR(R06),LS0PFLE IN USE AS ECB         HILOAD 00279000
         BNO   NEXT                NO - BRANCH                  HILOAD 00280000
         SPACE 1                                                HILOAD 00281000
         L     R15,LS0PBA-LS0PTR(,R06) SU ADDRESS               HILOAD 00282000
         SH    R15,IW0DESS         GET ECB DESCRIPTOR ADDRESS   HILOAD 00283000
         SPACE 1                                                HILOAD 00284000
         DXCECBD REG=R15           USE R15 FOR ECB DESCRIPTOR   HILOAD 00285000
         SPACE 1                                                HILOAD 00286000
         MVC   ITMCRI(3,R04),CE3CRI    CRI                      HILOAD 00287000
         MVC   ITMRHN(1,R04),CE3RHN+3  RECORD HOLD COUNT        HILOAD 00288000
         MVC   ITMCHN(1,R04),CE3CHN+3  RESOURCE HOLD COUNT      HILOAD 00289000
         SPACE 1                                                HILOAD 00290000
         AH    R15,IW0ECBD         GET ECB ADDRESS              HILOAD 00291000
         EB0EB REG=R15                                          HILOAD 00292000
         ST    R15,ITMECB(R04)     STORE ECB ADDRESS            HILOAD 00293000
         SPACE 1                                                HILOAD 00294000
         MVC   ITMPRG(4,R04),CE1PGM-EB0EB(R15) PROGRAM NAME     HILOAD 00295000
         MVC   ITMMAC(2,R04),CE1REQ-EB0EB(R15) LAST MACRO       HILOAD 00296000
         SPACE 1                                                HILOAD 00297000
******** CHECK THAT WE ARE STILL LOOKING AT THE SAME ECB         HILOAD 00298000
         SPACE 1                                                HILOAD 00299000
         CLC   ITMTOD(4,R04),LS0TOD-LS0PTR(R06) SAME CREAT TIME HILOAD 00300000
         BNE   LOOP2               NO - REPEAT THE EXTRACT      HILOAD 00301000
         SPACE 1                                                HILOAD 00302000
NEXT     EQU   *                                                HILOAD 00303000
         LA    R04,UTIMITML(,R04)  POINT TO NEXT ITEM           HILOAD 00304000
         BCTR  R05,0               DECREMENT REMAINING ITEMS    HILOAD 00305000
         SPACE 1                                                HILOAD 00306000
         LA    R06,LS1PTRL(,R06)   POINT TO NEXT DESCRIPTOR     HILOAD 00307000
```

```
         BCTR  RGF,0               DECREMENT REMAINING DESCRIP   HILOAD 00308000
         SPACE 1                                                 HILOAD 00309000
         LTR   R05,R05             LAST ITEM                     HILOAD 00310000
         BZ    WRITE               YES - BRANCH                  HILOAD 00311000
         SPACE 1                                                 HILOAD 00312000
         LTR   RGF,RGF             LAST DESCRIPTOR               HILOAD 00313000
         BZ    WRITE               YES - BRANCH                  HILOAD 00314000
         SPACE 1                                                 HILOAD 00315000
         B     LOOP2                                             HILOAD 00316000
         EJECT ,                                                 HILOAD 00317000
*------------------------------------------------------------    HILOAD 00318000
*        WRITE TO DIAGNOSTIC FILE                                HILOAD 00319000
*------------------------------------------------------------    HILOAD 00320000
         SPACE 1                                                 HILOAD 00321000
WRITE    EQU   *                                                 HILOAD 00322000
         LA    R14,USRSAV+16(R13) START OF DATA FOR TPEDMP       FIX001 00323000
         LA    R15,UTIMHDRL+(UTIMITML*UTIMITMC) LENGTH TO DUMP   HILOAD 00324000
         ST    R15,STORSIZE        LOAD LENGTH TO BE WRITTEN     HILOAD 00325000
         MVI   STORIND,IW0DIA      SET DATA DIAGNOSTIC FILE      HILOAD 00326000
         DXCSERV UWSEQ,PARM=((R14),STORSIZE,STORIND)             HILOAD 00327000
         SPACE 1                                                 HILOAD 00328000
         LTR   RGF,RGF             ANY MORE DESCRIPTORS          HILOAD 00329000
         BZ    UTIM2800            NO - BRANCH RETURN            HILOAD 00330000
         SPACE 1                                                 HILOAD 00331000
******** CLEAR OUTPUT AREA BEFORE GOING BACK FOR MORE DESCRIPTR  HILOAD 00332000
         SPACE 1                                                 HILOAD 00333000
         XC    USRSAV+16(256-16,R13),USRSAV+16(R13)              FIX001 00334000
         XC    USRSAV+256(256-USRSAV,R13),USRSAV+256(R13)        FIX001 00335000
         SPACE 1                                                 HILOAD 00336000
         MVC   UTIMID+USRSAV(3,R13),=C'EIO' OVERFLOW BLOCK       FIX001 00337000
         B     LOOP1               GO PROCESS REMAINING DESCR    HILOAD 00338000
         SPACE 1                                                 HILOAD 00339000
         DROP  R15                                               HILOAD 00340000
UTIM2800 EQU   *                                                 HILOAD 00341000
         SR    R15,R15             SET NO CHANGE RETURN CODE     HILOAD 00342000
         B     UTIM2900            RETURN                        HILOAD 00343000
         DROP  R14                                               HILOAD 00344000
         EJECT ,                                                 HILOAD 00345000
*================================================================* 00346000
*                                                                * 00347000
*        END OF USER CODE                                        * 00348000
*                                                                * 00349000
*================================================================* 00350000
         SPACE 1                                                   00351000
UTIM2900 DC    0H'0'                                               00352000
         DXCSAVE POP,RESTORE,      RESTORE CALLERS REGISTERS      -00353000
               EXCEPT=((R15))      (EXCEPT RETURN INFORMATION)     00354000
         BR    R14                 RETURN TO ALCS MONITOR          00355000
         EJECT ,                                                   00356000
*================================================================* 00357000
*                                                                * 00358000
*        START OF USER CONSTANTS                                 * 00359000
*                                                                * 00360000
*================================================================* 00361000
         SPACE 1                                                   00362000
         DS    0F                                              HILOAD 00363000
OUTMSG   DC    H'50'          LENGTH OF MSG                    HILOAD 00364000
         DC    H'0'                                            HILOAD 00365000
         DC    CL3'DXC'       LENGTH COUNT STARTS HERE         HILOAD 00366000
         DC    CL5'3000I'     MESSAGE IDENTIFIER               HILOAD 00367000
         DC    CL1' '                                          HILOAD 00368000
         DC    CL1' '                                          HILOAD 00369000
         DC    CL40'ALCS HIGH LOAD - ALCS CONTINUES        '   HILOAD 00370000
```

```
         SPACE 1                                                    HILOAD 00371000
*========================================================================* 00372000
*                                                                        * 00373000
*        END OF USER CONSTANTS                                           * 00374000
*                                                                        * 00375000
*========================================================================* 00376000
         EJECT ,                                                          00377000
         LTORG ,                                                          00378000
         DXCUEND ,                                                        00379000
         END                                                              00380000
```

HILOAD IW0USR

```
         MACRO ,                                                        00001000
&LABEL   IW0USR                                                         00002000
.*                                                                      00003000
.***********************************************************************00004000
.*       MACRO CREATED FOR HIGH LOAD MONITOR IMPLEMENTATION          *  00005000
.***********************************************************************00006000
.*                                                                      00007000
.* THIS MACRO DEFINES FIELDS IN THE MONITOR EXIT INFORMATION TABLE      00008000
.* USER AREA. THE TABLE IS DEFINED IN MACRO IW0IT, AND THE USER AREA    00009000
.* IS DEFINED AT LABEL IW0USER.                                         00010000
.*                                                                      00011000
.* THERE IS ONE FIELD CALLED "IW0HITOD" DEFINED IN THIS MACRO. THIS     00012000
.* FIELD HOLDS THE TIME WHEN THE LAST WARNING MESSAGE WAS SENT TO       00013000
.* THE SYSTEM CONSOLE.                                                  00014000
.*                                                                      00015000
         GBLB  &IW0USR             GLOBAL TO STOP DOUBLE GENERATION     00016000
         AIF   (&IW0USR).DONE      IF PREVIOUSLY CALLED, BRANCH         00017000
&IW0USR  SETB  (1)                 SET ON TO PREVENT DOUBLE GENERATION  00018000
         PRINT GEN                                                      00019000
IW0IT    DSECT ,                                                        00020000
         ORG   IW0USER                                                  00021000
IW0HITOD DS    F                   TIME WHEN LAST WARNING MSG SENT      00022000
&SYSECT  CSECT ,                                                        00023000
         PRINT NOGEN                                                    00024000
.DONE    MEXIT ,                                                        00025000
         MEND  ,                                                        00026000
```

HILOAD OFFLINE

```
*--------------------------------------------------------------------*  00001000
*                                                                    *  00002000
*     OFFLINE REPORT PROGRAM FOR 'MONITORING HIGH LOAD' FUNCTION     *  00003000
*                                                                    *  00004000
*--------------------------------------------------------------------*  00005000
*                                                                    *  00005100
*     PROGRAM UPDATED IN JANUARY 2001 (IMPROVED MACRO TABLE)         *  00005200
*                                                                    *  00005300
*--------------------------------------------------------------------*  00005400
         SPACE 1                                                       00006000
         PRINT NOGEN                                                   00007000
         REGEQ ,                                                       00008000
HILOAD   CSECT ,                                                       00009000
         SAVE  (14,12)             SAVE CALLERS REGISTERS              00010000
         LR    R8,R15              SET UP BASE REGISTER                00011000
         USING HILOAD,R8                                               00012000
         LR    R9,R13              SAVE CALLERS SAVE AREA ADDR         00013000
         LA    R13,SAVE1           SET UP OUR SAVEAREA                 00014000
         ST    R9,4(,R13)                                              00015000
         SPACE 1                                                       00016000
******** OPEN INFILE (DIAGNOSTIC FILE) AND OUTFILE (PRINT FILE)        00017000
         SPACE 1                                                       00018000
         OPEN  (INFILE,(INPUT))                                        00019000
         OPEN  (OUTFILE,(OUTPUT))                                      00020000
         SPACE 1                                                       00021000
******** LOOP TO HERE FOR EACH NEW RECORD FROM DIAG FILE               00022000
         SPACE 1                                                       00023000
NEWREC   EQU   *                                                       00024000
         GET   INFILE,INAREA                                           00025000
         SPACE 1                                                       00026000
         CLC   INAREA(2),=C'EI'    REQUIRED RECORDS FOR ANALYSIS       00027000
         BNE   NEWREC                                                  00028000
         SPACE 1                                                       00029000
         CLI   INAREA+2,C'P'       FIRST RECORD OF GROUP               00030000
         BNE   LAB001              NO - SKIP HEADER PROCESSING         00031000
         SPACE 1                                                       00032000
         MVC   TIME(4),INAREA+40             STORE TIME                00033000
         MVI   OUTAREA,X'40'                 CTL CHAR FOR SPACE 1 LINE 00034000
         MVC   OUTAREA+1(132),OUTAREA        CLEAR TO BLANKS           00035000
         PUT   OUTFILE,OUTAREA               PRINT BLANK LINE          00036000
         MVI   OUTAREA,X'40'                 CTL CHAR FOR SPACE 1 LINE 00037000
         MVI   OUTAREA+1,C'*'                SET TO *                  00038000
         MVC   OUTAREA+2(80),OUTAREA+1       SET TO *                  00039000
         PUT   OUTFILE,OUTAREA               PRINT LINE OF *           00040000
         MVI   OUTAREA,X'40'                 CTL CHAR FOR SPACE 1 LINE 00041000
         MVC   OUTAREA+1(132),OUTAREA        CLEAR TO BLANKS           00042000
         PUT   OUTFILE,OUTAREA               PRINT BLANK LINE          00043000
         MVI   OUTAREA,X'40'                 CTL CHAR FOR SPACE 1 LINE 00044000
         MVC   OUTAREA+1(132),OUTAREA CLEAR PRINT LINE TO BLANKS       00045000
         MVC   OUTAREA+1(HDR1LEN),HDR1                                 00046000
         PUT   OUTFILE,OUTAREA               PRINT FIRST HEADER LINE   00047000
         UNPK  WORKAREA(9),INAREA(5)                                   00048000
         UNPK  WORKAREA+8(9),INAREA+4(5)                               00049000
         UNPK  WORKAREA+16(9),INAREA+8(5)                              00050000
         UNPK  WORKAREA+24(9),INAREA+12(5)                             00051000
         UNPK  WORKAREA+32(9),INAREA+16(5)                             00052000
         UNPK  WORKAREA+40(9),INAREA+20(5)                             00053000
         UNPK  WORKAREA+48(9),INAREA+24(5)                             00054000
         UNPK  WORKAREA+56(9),INAREA+28(5)                             00055000
         UNPK  WORKAREA+64(9),INAREA+32(5)                             00056000
```

```
         UNPK  WORKAREA+72(9),INAREA+36(5)                              00057000
         UNPK  WORKAREA+80(9),INAREA+40(5)                              00058000
         TR    WORKAREA(88),TRTAB                                       00059000
         MVI   OUTAREA,X'40'               CTL CHAR FOR SPACE 1 LINE    00060000
         MVC   OUTAREA+1(132),OUTAREA      CLEAR TO BLANKS              00061000
         BAL   R7,SUBRTN                   DATE/TIME TO PRINTER         00062000
         MVC   OUTAREA+20(4),WORKAREA+8    ACTIVE ECB COUNT TO PRINT    00063000
         MVC   OUTAREA+25(4),WORKAREA+12   DELAY DEFER COUNT TO PRINT   00064000
         MVC   OUTAREA+30(4),WORKAREA+20   SUS AVAILABLE TO PRINT       00065000
         MVC   OUTAREA+35(4),WORKAREA+28   IOCBS AVAILABLE TO PRINT     00066000
         MVC   OUTAREA+40(4),WORKAREA+36   ACV1 TO PRINT                00067000
         MVC   OUTAREA+45(4),WORKAREA+44   ACV2 TO PRINT                00068000
         MVC   OUTAREA+50(4),WORKAREA+52   ACV3 TO PRINT                00069000
         MVC   OUTAREA+55(4),WORKAREA+60   ACV4 TO PRINT                00070000
         MVC   OUTAREA+60(4),WORKAREA+68   ACV5 TO PRINT                00071000
         MVC   OUTAREA+65(4),WORKAREA+76   ACV6 TO PRINT                00072000
         PUT   OUTFILE,OUTAREA                                          00073000
         MVI   OUTAREA,X'40'               CTL CHAR FOR SPACE 1 LINE    00074000
         MVC   OUTAREA+1(132),OUTAREA      CLEAR TO BLANKS              00075000
         PUT   OUTFILE,OUTAREA             BLANK LINE                   00076000
         MVI   OUTAREA,X'40'               CTL CHAR FOR SPACE 1 LINE    00077000
         MVC   OUTAREA+1(132),OUTAREA      CLEAR TO BLANKS              00078000
         MVC   OUTAREA+1(HDR2LEN),HDR2                                  00079000
         PUT   OUTFILE,OUTAREA             PRINT SECOND HEADER LINE     00080000
         SPACE 1                                                        00081000
LAB001   EQU   *                                                        00082000
         LA    R7,18               LOOP COUNT (ITEMS IN RECORD)         00083000
         LA    RGA,INAREA+48       START OF FIRST ITEM                  00084000
         SPACE 1                                                        00085000
******** LOOP TO HERE FOR EACH NEW ITEM                                  00086000
         SPACE 1                                                        00087000
NEWITEM  EQU   *                                                        00088000
         TM    0(RGA),X'C0'        SU IN USE                            00089000
         BNO   BUMP                                                     00090000
         SPACE 1                                                        00091000
         UNPK  WORKAREA(9),0(5,RGA)                                     00092000
         UNPK  WORKAREA+8(9),4(5,RGA)                                   00093000
         UNPK  WORKAREA+16(9),8(5,RGA)                                  00094000
         UNPK  WORKAREA+24(9),12(5,RGA)                                 00095000
         UNPK  WORKAREA+32(9),16(5,RGA)                                 00096000
         TR    WORKAREA(40),TRTAB                                       00097000
         SPACE 1                                                        00098000
         MVI   OUTAREA,X'40'       MOVE ASA CONTROL CHAR (SPACE 1)      00099000
         MVC   OUTAREA+1(132),OUTAREA CLEAR PRINT LINE TO BLANKS        00100000
         TM    0(RGA),X'E0'        ECB WAITING FOR INPUT MSG            00101000
         BNO   LAB010              NO - BRANCH                          00102000
         SPACE 1                                                        00103000
         MVC   OUTAREA+8(8),WORKAREA+16 MOVE ECB ADDR TO PRINT LINE     00104000
         MVC   OUTAREA+48(25),=C'ECB WAITING FOR INPUT MSG'             00105000
         B     PRINT                                                    00106000
         SPACE 1                                                        00107000
LAB010   EQU   *                                                        00108000
         MVC   OUTAREA+1(6),WORKAREA+2   MOVE CRI TO PRINT LINE         00109000
         MVC   OUTAREA+8(8),WORKAREA+16  MOVE ECB ADDR TO PRINT LINE    00110000
         MVC   OUTAREA+17(4),12(RGA)     MOVE PROGRAM NAME TO PRINT     00111000
         SR    R6,R6                     CLEAR FOR INSERT               00112000
         IC    R6,5(,RGA)                GET MACRO NUMBER               00113000
         SRL   R6,1                      ENSURE EVEN NUMBER             00114000
         SLL   R6,3                      MACRO NUMBER * 4               00115000
         L     R5,=A(MACTAB)             ADDRESS OF MACRO NAME TABLE    00116000
         AR    R6,R5                     POINT TO MACRO NAME            00117000
         MVC   OUTAREA+22(5),0(R6)       MOVE LAST MACRO TO PRINT LINE  00118000
         MVC   OUTAREA+28(2),WORKAREA+12 MOVE REC HELD TO PRINT LINE    00119000
```

```
         MVC   OUTAREA+31(2),WORKAREA+14 MOVE RES HELD TO PRINT LINE    00120000
         L     R6,TIME                   TIME THIS REPORT               00121000
         S     R6,16(,RGA)               LESS TIME ECB CREATED          00122000
         ST    R6,WORKAR2                SAVE ECB AGE IN SECONDS        00123000
         UNPK  WORKAR3(5),WORKAR2+2(3)   UNPACK ECB AGE IN SECONDS      00124000
         TR    WORKAR3(4),TRTAB          TRANSLATE FOR PRINTING         00125000
         MVC   OUTAREA+34(4),WORKAR3     MOVE ECB AGE TO PRINT          00126000
         SPACE 1                                                        00127000
PRINT    EQU   *                                                       00128000
         PUT   OUTFILE,OUTAREA                                         00129000
         SPACE 1                                                       00130000
BUMP     EQU   *                                                       00131000
         LA    RGA,20(,RGA)        BUMP TO NEXT ITEM                   00132000
         BCT   R7,NEWITEM          GO HANDLE NEW ITEM                  00133000
         SPACE 1                                                       00134000
         B     NEWREC              GO HANDLE NEW RECORD                00135000
         SPACE 1                                                       00136000
******** COME HERE AT END OF INPUT FILE                                 00137000
         SPACE 1                                                       00138000
ENDFILE  EQU   *                                                       00139000
         CLOSE (INFILE)                                                00140000
         CLOSE (OUTFILE)                                               00141000
         L     R13,4(,R13)         RESTORE CALLERS SAVEAREA            00142000
         RETURN (14,12),,RC=0      RETURN TO CALLER                    00143000
         EJECT ,                                                       00144000
******** SUBROUTINE TO PUT DATE/TIME INTO PRINT LINE                    00145000
*        THIS SUBROUTINE TAKES NO ACCOUNT OF LOCAL/GMT VARIATION       00146000
         SPACE 1                                                       00147000
SUBRTN   EQU   *                                                       00148000
         L     R14,INAREA+40       TIME DATA DUMPED                    00149000
         SL    R14,=X'8F809FD3'    RESET ORIGIN TO 1980                00150000
         SR    R15,R15             CLEAR FOR DIVIDE                    00151000
         SRDL  R14,12              CONVERT TO MICROSECONDS             00152000
         D     R14,=F'60000000'    MINUTES IN R15                     -00153000
                                   MICROSECONDS REMAINDER IN R14       00154000
         LR    R01,R14             MICROSECONDS REMAINDER              00155000
         SR    R0,R0               CLEAR FOR DIVIDE                    00156000
         D     R0,=F'1000000'      SECONDS IN R1                       00157000
         CVD   R01,WORKAREA+88                                         00158000
         UNPK  OUTAREA+14(2),WORKAREA+88(8) SECONDS TO PRINT LINE      00159000
         OI    OUTAREA+15,X'F0'                                        00160000
         SR    R14,R14             CLEAR FOR DIVIDE                    00161000
         D     R14,=F'60'          HOURS IN R15                       -00162000
                                   MINUTES REMAINDER IN R14            00163000
         CVD   R14,WORKAREA+88                                         00164000
         UNPK  OUTAREA+11(2),WORKAREA+88(8) MINUTES TO PRINT LINE      00165000
         OI    OUTAREA+12,X'F0'                                        00166000
         SR    R14,R14             CLEAR FOR DIVIDE                    00167000
         D     R14,=F'24'          DAYS IN R15                        -00168000
                                   HOURS REMAINDER IN R14              00169000
         CVD   R14,WORKAREA+88                                         00170000
         UNPK  OUTAREA+8(2),WORKAREA+88(8) HOURS TO PRINT LINE         00171000
         OI    OUTAREA+9,X'F0'                                         00172000
         SR    R14,R14             CLEAR FOR DIVIDE                    00173000
         D     R14,=A((365*4)+1)   DIVIDE BY DAYS IN 4 YEARS          -00174000
                                   QUOTIENT IN R15                    -00175000
                                   REMAINDER IN R14                    00176000
         AR    R15,R15             MULTIPLY BY 4 AND                   00177000
         LA    R15,80(R15,R15)     ADD BASE YEAR                       00178000
         LA    R01,366             LOAD DAYS IN LEAP YEAR              00179000
         SR    R14,R1              DECREMENT DAYS IN YEAR 1            00180000
         BM    SUBRTN1             BRANCH IF IN YEAR 1                 00181000
         SPACE 1                                                       00182000
```

```
         LA    R15,1(,R15)         INCREMENT YEAR                        00183000
         LA    R01,365             LOAD DAYS IN YEAR 2 3 OR 4            00184000
         SR    R14,R1              DECREMENT DAYS IN YEAR 2              00185000
         BM    SUBRTN1             BRANCH IF IN YEAR 2                   00186000
         SPACE 1                                                         00187000
         LA    R15,1(,R15)         INCREMENT YEAR                        00188000
         SR    R14,R1              DECREMENT DAYS IN YEAR 3              00189000
         BM    SUBRTN1             BRANCH IF IN YEAR 3                   00190000
         SPACE 1                                                         00191000
         LA    R15,1(,R15)         INCREMENT YEAR                        00192000
         SR    R14,R1              DECREMENT DAYS IN YEAR 4              00193000
         SPACE 1                                                         00194000
SUBRTN1  EQU   *                                                         00195000
         AR    R14,R1              RESTORE DAYS REMOVED BEFORE BRANCH    00196000
         LA    R14,1(,R14)         JAN1 IS DAY 1 (NOT DAY 0)             00197000
         CVD   R14,WORKAREA+88                                           00198000
         UNPK  OUTAREA+4(3),WORKAREA+88(8) DAYS TO PRINT LINE            00199000
         OI    OUTAREA+6,X'F0'                                           00200000
         CVD   R15,WORKAREA+88                                           00201000
         UNPK  OUTAREA+1(2),WORKAREA+88(8) YEARS TO PRINT LINE           00202000
         OI    OUTAREA+2,X'F0'                                           00203000
         MVI   OUTAREA+3,C'.'                                            00204000
         MVI   OUTAREA+10,C'.'                                           00205000
         MVI   OUTAREA+13,C'.'                                           00206000
         BR    R07                 RETURN TO MAINLINE                    00207000
         EJECT ,                                                         00208000
******** DCB ABEND EXIT ROUTINE FOR INPUT                                 00209000
         SPACE 1                                                         00210000
INPUTER  EQU   *                                                         00211000
         MVI   3(R01),4            SET TO IGNORE ERROR                   00212000
         RETURN ,                  AND RETURN TO CALLER                  00213000
         SPACE 1                                                         00214000
         DS    0D                                                        00215000
         DC    CL8'DCBIN   '       EYECATCH FOR DUMPS                    00216000
INFILE   DCB   DDNAME=FILEIN,DSORG=PS,MACRF=GD,RECFM=VBS,               -00217000
               EODAD=ENDFILE,EROPT=SKP,EXLST=INPUTXL                     00218000
         SPACE 1                                                         00219000
INPUTXL  DC    AL1(X'11'+X'80'),AL3(INPUTER)                             00220000
         SPACE 1                                                         00221000
         DS    0D                                                        00222000
         DC    CL8'DCBOUT  '       EYECATCH FOR DUMPS                    00223000
OUTFILE  DCB   DDNAME=FILEOUT,DSORG=PS,MACRF=PM,RECFM=FBA,              -00224000
               LRECL=133,BLKSIZE=133                                     00225000
         SPACE 1                                                         00226000
         DS    0D                                                        00227000
         SPACE 1                                                         00228000
         LTORG ,                                                         00229000
         SPACE 1                                                         00230000
         DS    0D                                                        00231000
WORKAREA DS    104C                                                      00232000
WORKAR2  DS    8C                                                        00233000
WORKAR3  DS    8C                                                        00234000
TIME     DC    F'0'                                                      00235000
TRTAB    EQU   *-240                                                     00236000
         DC    C'0123456789ABCDEF'                                       00237000
HDR1     DC    C'DATE   TIME        ACT  DEL  SU   IOCB '               00238000
         DC    C'ACV1 ACV2 ACV3 ACV4 ACV5 ACV6'                          00239000
HDR1LEN  EQU   *-HDR1                                                    00240000
HDR2     DC    C'CRI    ECB      PROG MACRO -HLD- AGE'                   00241000
HDR2LEN  EQU   *-HDR2                                                    00242000
         DS    0D                                                        00243000
         DC    CL8'SAVEAREA'       EYECATCH FOR DUMPS                    00244000
SAVE1    DC    9D'0'               OUR SAVE AREA                         00245000
```

```
         SPACE 1                                                        00246000
         DS    0D                                                       00247000
         DC    CL8'OUTAREA '       EYECATCH FOR DUMPS                   00248000
OUTAREA  DS    133C                OUTPUT AREA                          00249000
         DS    0D                                                       00250000
         DC    CL8'INAREA  '       EYECATCH FOR DUMPS                   00251000
INAREA   DS    13000C              INPUT AREA                           00252000
         DS    0D                                                       00253000
MACTAB   DC    CL8'        '       0                                    00254000
         DC    CL8'BACKC   '       2                                    00255000
         DC    CL8'CREDC   '       4                                    00256000
         DC    CL8'CREMC   '       6                                    00257000
         DC    CL8'CRETCM  '       8                                    00258000
         DC    CL8'DEFRC   '       10                                   00259000
         DC    CL8'DLAYC   '       12                                   00260000
         DC    CL8'ENTDC   '       14                                   00261000
         DC    CL8'ENTNC   '       16                                   00262000
         DC    CL8'ENTRC   '       18                                   00263000
         DC    CL8'EXITC   '       20                                   00264000
         DC    CL8'FILEC   '       22                                   00265000
         DC    CL8'FLNPC   '       24                                   00266000
         DC    CL8'PLEXC   '       26                                   00267000
         DC    CL8'FILUC   '       28                                   00268000
         DC    CL8'FINDC   '       30                                   00269000
         DC    CL8'FINHC   '       32                                   00270000
         DC    CL8'FNDPC   '       34                                   00271000
         DC    CL8'APPCC   '       36                                   00272000
         DC    CL8'FINWC   '       38                                   00273000
         DC    CL8'FIWHC   '       40                                   00274000
         DC    CL8'FLIPC   '       42                                   00275000
         DC    CL8'GETCC   '       44                                   00276000
         DC    CL8'BLKIC   '       46                                   00277000
         DC    CL8'HLLCC   '       48                                   00278000
         DC    CL8'RELCC   '       50                                   00279000
         DC    CL8'RLCHC   '       52                                   00280000
         DC    CL8'RELFC   '       54                                   00281000
         DC    CL8'SENDCA  '       56                                   00282000
         DC    CL8'WTOPC   '       58                                   00283000
         DC    CL8'ROUTC   '       60                                   00284000
         DC    CL8'SENDCD  '       62                                   00285000
         DC    CL8'SYNCC   '       64                                   00286000
         DC    CL8'TCLSC   '       66                                   00287000
         DC    CL8'TOPNC   '       68                                   00288000
         DC    CL8'TASNC   '       70                                   00289000
         DC    CL8'TPRDC   '       72                                   00290000
         DC    CL8'TSQCC   '       74                                   00291000
         DC    CL8'TRSVC   '       76                                   00292000
         DC    CL8'TWRTC   '       78                                   00293000
         DC    CL8'UNFRC   '       80                                   00294000
         DC    CL8'MONTC   '       82                                   00295000
         DC    CL8'LMONC   '       84                                   00296000
         DC    CL8'HASHC   '       86                                   00297000
         DC    CL8'TDTAC   '       88                                   00298000
         DC    CL8'TDSPC   '       90                                   00299000
         DC    CL8'TASTC   '       92                                   00300000
         DC    CL8'TASBC   '       94                                   00301000
         DC    CL8'CINFC   '       96                                   00302000
         DC    CL8'FILNC   '       98                                   00303000
         DC    CL8'CRETCS  '       100                                  00304000
         DC    CL8'TOURC   '       102                                  00305000
         DC    CL8'CPICC   '       104                                  00306000
         DC    CL8'CSQLC   '       106                                  00307000
         DC    CL8'SPOCC   '       108                                  00308000
```

```
DC    CL8'GGFAC   '       110                                   00309000
DC    CL8'SENDCT  '       112                                   00310000
DC    CL8'CREXC   '       114                                   00311000
DC    CL8'KEYCC   '       116                                   00312000
DC    CL8'KEYRC   '       118                                   00313000
DC    CL8'GFSCC   '       120                                   00314000
DC    CL8'SXIPC   '       122                                   00315000
DC    CL8'CMQIC   '       124                                   00316000
DC    CL8'TOUTC   '       126                                   00317000
DC    CL8'KEYUC   '       128                                   00318000
DC    CL8'TIMEC   '       130                                   00319000
DC    CL8'SOCKC   '       132                                   00320000
DC    CL8'HELPC   '       134                                   00321000
DC    CL8'SAVEC   '       136                                   00322000
DC    CL8'ALASC   '       138                                   00323000
DC    CL8'STXTC   '       140                                   00324000
DC    CL8'SEOMC   '       142                                   00325000
DC    CL8'SANSC   '       144                                   00326000
DC    CL8'SCDCCS  '       146                                   00327000
DC    CL8'SCDCCM  '       148                                   00328000
DC    CL8'POLLC   '       150                                   00329000
DC    CL8'REQSC   '       152                                   00330000
DC    CL8'FINPC   '       154                                   00331000
DC    CL8'TPSWC   '       156                                   00332000
DC    CL8'SENDCM  '       158                                   00333000
DC    CL8'PURGC   '       160                                   00334000
DC    CL8'PLONC1  '       162                                   00335000
DC    CL8'PLONC2  '       164                                   00336000
DC    CL8'SENDCK  '       166                                   00337000
DC    CL8'KCTLC   '       168                                   00338000
DC    CL8'HALTC   '       170                                   00339000
DC    CL8'FACEC   '       172                                   00340000
DC    CL8'PFACC   '       174                                   00341000
DC    CL8'RAISC   '       176                                   00342000
DC    CL8'RONIC   '       178                                   00343000
DC    CL8'GETFC   '       180                                   00344000
DC    CL8'ATTAC   '       182                                   00345000
DC    CL8'DETAC   '       184                                   00346000
DC    CL8'CORHC   '       186                                   00347000
DC    CL8'CORUC   '       188                                   00348000
DC    CL8'WAITC   '       190                                   00349000
DC    CL8'GTFCC   '       192                                   00350000
DC    CL8'STVCC   '       194                                   00351000
DC    CL8'DCRCC   '       196                                   00352000
DC    CL8'DASCC   '       198                                   00353000
DC    CL8'FIPWC   '       200                                   00354000
DC    CL8'RIDIC   '       202                                   00355000
DC    CL8'SLIMC   '       204                                   00356000
DC    CL8'DMPCC   '       206                                   00357000
DC    CL8'ADRIC   '       208                                   00358000
DC    CL8'STICC   '       210                                   00359000
DC    CL8'SYSCC   '       212                                   00360000
DC    CL8'COMIC   '       214                                   00361000
DC    CL8'COMCC   '       216                                   00362000
DC    CL8'LODIC   '       218                                   00363000
DC    CL8'PGMCC   '       220                                   00364000
DC    CL8'ECBIC   '       222                                   00365000
DC    CL8'UNHKA   '       224                                   00366000
DC    CL8'REHKA   '       226                                   00367000
DC    CL8'WHOCC   '       228                                   00368000
DC    CL8'TDTAC   '       230                                   00369000
DC    CL8'SONIC   '       232                                   00370000
DC    CL8'SENDCS  '       234                                   00371000
```

```
DC    CL8'HLDTC   '       236                                   00372000
DC    CL8'CPDUC   '       238                                   00373000
DC    CL8'AUTHC   '       240                                   00374000
DC    CL8'COMCC   '       242                                   00375000
DC    CL8'ASCIC   '       244                                   00376000
DC    CL8'WILDC   '       246                                   00377000
DC    CL8'TDFAC   '       248                                   00378000
DC    CL8'TDFBC   '       250                                   00379000
DC    CL8'DECBC   '       252                                   00380000
DC    CL8'COMTC   '       254                                   00381000
DC    CL8'        '       256                                   00382000
END   ,                                                         00383000
```

End Of Document